TB 13.10 Cavitation 01-24-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

This is the 5th in a series of notebooks in which I work through material and exercises in the magisterial new book *Modern Classical Physics* by Kip S. Thorne and Roger D. Blandford. If you are a physicist of any ilk, BUY THIS BOOK. You will learn from a close reading and from solving the exercises.



Fig. 13.8: Water flowing past a hydrofoil as seen in the hydrofoil's rest frame.

33

Exercise 13.10 *Problem: Cavitation*
A hydrofoil moves with speed V at a depth $D = 3$m below the surface of a lake; see Figure 13.8. Estimate how fast V must be to make the water next to the hydrofoil boil. (This boiling, which is called *cavitation*, results from the pressure P trying to go negative.)

[Note: for a more accurate value of the speed V that triggers cavitation, one would have to compute the velocity field around the hydrofoil, e.g. using the method of Ex. 14.8 of the next chapter.]

Analysis and solution

This problem is a simple application of the invariance of the Bernoulli constant along a stream line. Cavitation will occur if the pressure at point B is less than the vapor pressure of water.



In this context, the Bernoulli constant is

$$B = \frac{V^2}{2} + h + \Phi = \frac{V^2}{2} + u + \frac{P}{\rho} + \Phi$$

```
w[1] = VA^2/2 + uA + PA/ρA + ΦA == VB^2/2 + uB + PB/ρB + ΦB
```

$$uA + \frac{VA^2}{2} + \frac{PA}{\rho A} + \Phi A == uB + \frac{VB^2}{2} + \frac{PB}{\rho B} + \Phi B$$

The water is incompressible and the internal energy of the water is unchanged near the hydrofoil

```
w[2] = w[1] /. uB → uA /. ρB → ρ /. ρA → ρ /. ΦA → ρ g d /. ΦB → ρ g (d - w/2)
```

$$uA + \frac{VA^2}{2} + \frac{PA}{\rho} + d\, g\, \rho == uA + \frac{VB^2}{2} + \frac{PB}{\rho} + g \left(d - \frac{w}{2}\right) \rho$$

where w is the assumed height of the hydrofoil. Away from the hydrofoil, the flow velocity is zero. Also, since the size of the hydrofoil was not provided in the problem statement, it can be assumed that w << d.

```
w[3] = w[2] /. VA → 0 /. w → 0
```

$$uA + \frac{PA}{\rho} + d\, g\, \rho == uA + \frac{VB^2}{2} + \frac{PB}{\rho} + d\, g\, \rho$$

```
w[4] = Solve[w[3], PB][[1, 1]]
```

$$PB \to \frac{1}{2} \left(2\, PA - VB^2 \rho\right)$$

Introduce the pressure at A

```
w[5] = w[4] /. PA → ρ g d /. Rule → Equal
```

$$PB == \frac{1}{2}\left(2\, d\, g\, \rho - VB^2\, \rho\right)$$

Cavitation occurs when PB becomes less than the vapor pressure of the water at point B

```
w[6] = Solve[w[5] /. PB → PV, VB]〚2, 1〛 /. Rule → Equal
```

$$VB == \frac{\sqrt{2}\,\sqrt{-PV + d\, g\, \rho}}{\sqrt{\rho}}$$

From the web I find https://en.wikipedia.org/wiki/Vapour_pressure_of_water

Vapour pressure of water (0–100 °C)[1]

T, °C	*T*, °F	*P*, kPa	*P*, torr	*P*, atm
0	32	0.6113	4.5851	0.0060
5	41	0.8726	6.5450	0.0086
10	50	1.2281	9.2115	0.0121
15	59	1.7056	12.7931	0.0168
20	68	2.3388	17.5424	0.0231
25	77	3.1690	23.7695	0.0313
30	86	4.2455	31.8439	0.0419
35	95	5.6267	42.2037	0.0555

and choose the representative value PV = 10 torr https://en.wikipedia.org/wiki/Torr#Conversion_factors

Pressure units

V·T·E	Pascal (Pa)	Bar (bar)	Technical atmosphere (at)	Standard atmosphere (atm)	Torr (Torr)	Pounds per square inch (psi)
1 Pa	≡ 1 N/m^2	10^{-5}	1.0197×10^{-5}	9.8692×10^{-6}	7.5006×10^{-3}	$1.450\,377 \times 10^{-4}$
1 bar	10^5	≡ 100 kPa ≡ 10^6 dyn/cm^2	1.0197	0.986 92	750.06	14.503 77
1 at	$9.806\,65 \times 10^4$	0.980 665	≡ 1 kp/cm^2	0.967 8411	735.5592	14.223 34
1 atm	$1.013\,25 \times 10^5$	1.013 25	1.0332	1	≡ 760	14.695 95
1 Torr	133.3224	$1.333\,224 \times 10^{-3}$	$1.359\,551 \times 10^{-3}$	≡ 1/760 ≈ $1.315\,789 \times 10^{-3}$	≡ 1 Torr ≈ 1 mmHg	$1.933\,678 \times 10^{-2}$
1 psi	6.8948×10^3	6.8948×10^{-2}	$7.030\,69 \times 10^{-2}$	6.8046×10^{-2}	51.714 93	≡ 1 lbf /in^2

```
Module[{g = 9.8 m/s^2, d = 3 m, ρ = 1 gm/cm^3, PV = 10 torr},
 VB = (Sqrt[2] Sqrt[-PV + d g ρ])/Sqrt[ρ] /.
   {gm → kg/1000, cm → m/100 , torr → 133 (nt/m^2) /. nt → kg m/s^2 };
 VB = Simplify[VB, {m > 0, kg > 0, s > 0}];
 {VB, VB /. {m → km/1000, s → hr/3600} /. km → 0.632 mi}]
```

$$\left\{\frac{7.49266\,\text{m}}{\text{s}}, \frac{17.0473\,\text{mi}}{\text{hr}}\right\}$$

Cavitation begins at approximately 17 mi/hr.

Visualization

```
Module[{xS = -5, xF = 5, d = -3, dLine, VVector, range, surface, ellipsoid, lab,
  streamlineArrows, markers, FStream, StreamlineMarker, StreamlineArrow, G},
 range = {{xS, xF}, {-5, 0}};
 dLine = {Gray, Line[{{0, 0}, {0, d}}], Text[Stl["d"], {-0.2, d/2}]};
 VVector = {Black, Arrow[{{0, d}, {-1, d}}] , Text[Stl["V"], {-0.5, d - 0.25}]};
 surface = {Directive[Blue, Thickness[0.03]], Line[{{xS, 0}, {xF, 0}}]};
 ellipsoid = {Gray, Ellipsoid[{0, d}, {2, 0.8}]};
 lab = Stl["Schematic representation of hydrofoil and a streamline"];

 (* Not a real streamline. Just a
  function with the approximate shape of a streamline. *)
 FStream[x_] := d + Sech[x]^(1/2);
 StreamlineArrow[x_, δ_] := Arrow[{{x, FStream[x]}, {x + δ, FStream[x + δ]}}];
 StreamlineMarker[lab_, x_, δ_] :=
  {Black, Point[{}], Text[Style[lab], {x, FStream[x]} + {δ, δ}]};
 streamlineArrows = {Arrowheads[0.05],
   StreamlineArrow[-3, 0.5], StreamlineArrow[3, 0.5]};
 markers = {Black, StreamlineMarker["A", -4.5, 0.2], StreamlineMarker["B", 0, 0.2]};
 G[1] =
  Plot[FStream[x], {x, xS, xF}, Axes → None, PlotRange → range, PlotLabel → lab];
 G[2] = Graphics[{surface, ellipsoid, dLine, markers, VVector, streamlineArrows},
   Axes → None, PlotRange → range];
 Show[G[1], G[2]]]
```



References

Following the references given in the statement of the problem, I found a good discussion of the physics of this problem in my pdf copy of Batchelor, p 481. In addition, I found an excellent discussion of the TB problem *13.7 Hole in my bucket,* Section 6.3.1. I note that the key to most of the problems in the section is the invocation of the constancy of the Bernoulli function along a streamline.

Videos demonstrating cavitation are
https://www.youtube.com/watch?v=ON_irzFAU9c
https://www.youtube.com/watch?v=U-uUYCFDTrc   Lots of detail
https://www.youtube.com/watch?v=K_w3gcvA87I Illustrations of hydrofoils
https://www.youtube.com/watch?v=K_w3gcvA87I   Delightful old video with several examples of cavitation and the underlying physics.

Caltech lecture
https://core.ac.uk/download/pdf/4893014.pdf

Recent thesis on cavitation   (lots of refs)
file:///C:/Users/NTG/Downloads/mscThesis_GJMeijn_220915%20(1).pdf

Text on cavitation
Brennen - Cavitation and Bubble Dynamics (I have the pdf)